

MCS®-51 INSTRUCTION SET

Table 10. 8051 Instruction Set Summary

Interrupt Response Time: Refer to Hardware Description Chapter.

Instructions that Affect Flag Settings(1)

Instruction		Flag	J	Instruction		Flag	I
	С	OV	AC		С	OV	AC
ADD	Х	Х	Х	CLR C	0		
ADDC	Х	X	Х	CPL C	Х		
SUBB	Х	Х	Х	ANL C,bit	Х		
MUL	0	Х		ANL C,/bit	Х		
DIV	0	Х		ORL C,bit	Х		
DA	Х			ORL C,bit	Х		
RRC	Х			MOV C,bit	Х		
RLC	Х			CJNE	Х		
SETB C	1						

(1)Note that operations on SFR byte address 208 or bit addresses 209-215 (i.e., the PSW or bits in the PSW) will also affect flag settings.

Note on instruction set and addressing modes:

Rn	— Register R7-R0 of the currently se-
	lected Register Bank.
direct	- 8-bit internal data location's address.
	This could be an Internal Data RAM
	location (0-127) or a SFR [i.e., I/O
	port, control register, status register, etc. (128-255)].
@Ri	— 8-bit internal data RAM location (0-
e Id	•
	255) addressed indirectly through reg-
	ister R1 or R0.
#data	 8-bit constant included in instruction.
#data 16	6 — 16-bit constant included in instruction.
addr 16	— 16-bit destination address. Used by
	LCALL & LJMP. A branch can be
	anywhere within the 64K-byte Pro-
	gram Memory address space.
addr 11	
auui II	•
	ACALL & AJMP. The branch will be
	within the same 2K-byte page of pro-

following instruction.

lowing instruction.

rel

bit

gram memory as the first byte of the

 Signed (two's complement) 8-bit offset byte. Used by SJMP and all conditional jumps. Range is -128 to +127 bytes relative to first byte of the fol-

Direct Addressed bit in Internal Data

RAM or Special Function Register.

Mne	monic	Description	Byte	Oscillator Period
ARITH	METIC OPI	RATIONS		
ADD	A,Rn	Add register to Accumulator	1	12
ADD	A,direct	Add direct byte to	2	12
ADD	A,@Ri	Accumulator Add indirect RAM	1	12
ADD	A,#data	to Accumulator Add immediate data to	2	12
ADDC	A,Rn	Accumulator Add register to Accumulator with Carry	1	12
ADDC	A,direct	Add direct byte to Accumulator	2	12
ADDC	A,@Ri	with Carry Add indirect RAM to Accumulator	1	12
ADDC	A,#data	with Carry Add immediate data to Acc with Carry	2	12
SUBB	A,Rn	Subtract Register from Acc with	1	12
SUBB	A,direct	Subtract direct byte from Acc with borrow	2	12
SUBB	A,@Ri	Subtract indirect RAM from ACC with borrow	1	12
SUBB	A,#data	Subtract immediate data from Acc with borrow	2	12
INC	A	Increment Accumulator	1	12
INC INC	Rn direct	Increment register Increment direct byte	1 2	12 12
INC	@Ri	Increment direct	1	12
DEC	Α	Decrement Accumulator	1	12
DEC	Rn	Decrement Register	1	12
DEC	direct	Decrement direct	2	12
DEC	@Ri	Decrement indirect RAM	1	12

All mnemonics copyrighted @Intel Corporation 1980



Table 10. 8051 Instruction Set Summary (Continued)

M	nemonic	Description	Byte	Oscillator Period
ARITI	HMETIC OPER	ATIONS (Continue	4)	
INC	DPTR	Increment Data	u, 1	24
	D	Pointer	•	27
MUL	AB	Multiply A & B	1	48
DIV	AB	Divide A by B	i	48
DA	A	Decimal Adjust	;	12
UA	^	Accumulator	1	12
LOGI	CAL OPERATI			
	A,Rn	AND Register to	1	12
ANL	д,пн	Accumulator	1	12
ANII	A,direct	AND direct byte	2	12
MILE	A,GII OCI	to Accumulator	2	12
ABII	A.@Ri		1	40
ANL	A,@HI	AND indirect	1	12
		RAM to		
		Accumulator	_	
ANL	A, # data	AND immediate	2	12
		data to		
		Accumulator		
ANL	direct,A	AND Accumulator	2	12
		to direct byte		
ANL	direct, # data	AND immediate	3	24
		data to direct byte		
ORL	A,Rn	OR register to	1	12
		Accumulator		
ORL	A, direct	OR direct byte to	2	12
		Accumulator		
ORL	A,@Ri	OR indirect RAM	1	12
		to Accumulator		
ORL	A, #data	OR immediate	2	12
	,	data to		
		Accumulator		
ORL	direct,A	OR Accumulator	2	12
		to direct byte	_	
ORL	direct. # data	OR immediate	3	24
		data to direct byte	-	
XRL	A Rn	Exclusive-OR	1	12
/\.	, 4,	register to	•	
		Accumulator		
VDI	A.direct	Exclusive-OR	2	12
ALL	A, all ect	direct byte to	~	12
		Accumulator		
VDI	A,@Ri	Exclusive-OR	1	12
ANL	A,eni	indirect RAM to	'	12
		Accumulator		
XRL	A.#data	Exclusive-OR	2	10
VLIF	n, #udiä		2	12
		immediate data to		
		Accumulator	_	
XRL	direct,A	Exclusive-OR	2	12
		Accumulator to		
l		direct byte		
XRL	direct, #data		3	24
		immediate data		
		to direct byte		
CLR	Α	Clear	1	12
		Accumulator		
CPL	Α	Complement	1	12
l		Accumulator		

Mr	nemonic	Description	Byte	Oscillator Period
LOGIC	AL OPERATIO	NS (Continued)		
RL	Α	Rotate	1	12
11.2	^	Accumulator Left	•	
RLC			4	12
HLU	A	Rotate	1	12
		Accumulator Left		
		through the Carry		
RR	Α	Rotate	1	12
		Accumulator		
		Right		
RRC	Α	Rotate	1	12
	• •	Accumulator	•	•
		Right through		
		the Carry		
SWAP	A	Swap nibbles	1	12
		within the		
		Accumulator		
DATA	TRANSFER			
MOV		Move	1	12
	,	register to	•	
		•		
		Accumulator	_	
MOV	A, direct	Move direct	2	12
		byte to		
		Accumulator		
MOV	A,@Ri	Move indirect	1	12
	•	RAM to		
		Accumulator		
MOV	A # data		2	12
MOV	A, # data	Move	2	12
		immediate		
İ		data to		
l		Accumulator		
MOV	Rn,A	Move	1	12
]		Accumulator		
		to register		
MOV	Rn,direct	•	2	24
IVICV	mii,uiieçi	Move direct	2	24
		byte to		
		register		
MOV	Rn,#data	Move	2	12
		immediate data		
		to register		
MOV	direct.A	Move	2	12
"""	an oct,r	Accumulator	-	
моч	diament Da	to direct byte	_	
MOV	direct,Rn	Move register	2	24
		to direct byte		
MOV	direct,direct	Move direct	3	24
1		byte to direct		
MOV	direct,@Ri	Move indirect	2	24
		RAM to	-	
1				
		direct byte	_	
MOV	direct, # data		3	24
1		immediate data		
1		to direct byte		
MOV	@Ri,A	Move	1	12
l	•	Accumulator to		
		indirect RAM		
1		HOHECL NAM		

All mnemonics copyrighted © Intel Corporation 1980



Table 10. 8051 Instruction Set Summary (Continued)

<u> </u>				Oscillator
, A	Anemonic	Description	Byte	Period
	TRANSFER (Con	tinued)		
MOV	@Ri,direct	Move direct	2	24
		byte to		
		indirect RAM		
MOV	@Ri,#data	Move	2	12
		immediate		
		data to		
	DDTD # 4-4-40	indirect RAM	_	•
MOV	DPTR,#data16		3	24
		Pointer with a		
MOVC	A.@A+DPTR	16-bit constant Move Code		24
MOVC	A, WATUFIN	byte relative to	1	24
		DPTR to Acc		
MOVC	A,@A+PC	Move Code	1	24
1.4000	A, WATEU	byte relative to	•	24
		PC to Acc		
моух	A.@Ri	Move	1	24
	,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,	External	•	
		RAM (8-bit		
		addr) to Acc		
MOVX	A,@DPTR	Move	1	24
	. ,	External	-	
		RAM (16-bit		
		addr) to Acc		
MOVX	@Ri,A	Move Acc to	1	24
		External RAM		
		(8-bit addr)		
MOVX	@DPTR,A	Move Acc to	1	24
		External RAM		
		(16-bit addr)		
PUSH	direct	Push direct	2	24
		byte onto		
		stack		
POP	direct	Pop direct	2	24
		byte from		
VO	4.5-	stack		40
XCH	A,Rn	Exchange	1	12
		register with		
хсн	A,direct	Accumulator	•	40
IVOH	A,OHECE	Exchange direct byte	2	12
1		direct byte with		
		Accumulator		
хсн	A,@Ri	Exchange	1	12
	, ,, 4, 11	indirect RAM	•	142
1		with		
		Accumulator		
XCHD	A,@Ri	Exchange low-	1	12
		order Digit	•	
		indirect RAM		
		with Acc		

Mner	nonic	Description	Byte	Oscillator Period
BOOLE	AN VARIA	BLE MANIPULATION	ON	
CLR	С	Clear Carry	1	12
CLR	bit	Clear direct bit	2	12
SETB	C	Set Carry	1	12
SETB	bit	Set direct bit	2	12
CPL	С	Complement Carry	1	12
CPL	bit	Complement direct bit	2	12
ANL	C,bit	AND direct bit to CARRY	2	24
ANL	C,/bit	AND complement of direct bit to Carry	2	24
ORL	C,bit	OR direct bit to Carry	2	24
ORL	C,/bit	OR complement of direct bit to Carry	2	24
MOV	C,bit	Move direct bit to Carry	2	12
MOV	bit,C	Move Carry to	2	24
1C	rel	Jump if Carry	2	24
JNC	rel	Jump if Carry not set	2	24
JB	bit,rel	Jump if direct Bit is set	3	24
JNB	bit,rel	Jump if direct Bit is Not set	3	24
JBC	bit,rel	Jump if direct Bit is set & clear bit	3	24
PROGR	AM BRAN			
ACALL		Absolute Subroutine Call	2	24
LCALL	addr16	Long Subroutine Call	3	24
RET		Return from Subroutine	1	24
RETI		Return from interrupt	1	24
AJMP	addr11	Absolute Jump	2	24
LJMP SJMP	addr16 rel	Long Jump Short Jump (relative addr)	3 2	24 24

All mnemonics copyrighted © Intel Corporation 1980



Table 10. 8051 Instruction Set Summary (Continued)

Mr	nemonic	Description	Byte	Oscillator Period
PROGI	RAM BRANCH	IING (Continued)		
JMP	@A+DPTR	Jump indirect relative to the	1	24
•		DPTR		
JΖ	rel	Jump if	2	24
		Accumulator		
		is Zero		
JNZ	rel	Jump if	2	24
		Accumulator		
		is Not Zero	_	
CUNE	A,direct,rel	Compare	3	24
		direct byte to		
i		Acc and Jump		
		if Not Equal		
CINE	A,#data,rel	Compare	3	24
		immediate to		
		Acc and Jump		
1		if Not Equal		

M	nemonic	Description	Byte	Oscillator Period
PROGI	RAM BRANCHII	NG (Continued)		
CJNE	Rn, # data,rel	Compare immediate to register and Jump if Not	3	24
		Equal		
CJNE	@Ri, # data,rel	Compare immediate to indirect and Jump if Not Equal	3	24
DJNZ	Rn,rel	Decrement register and Jump if Not Zero	2	24
DJNZ	direct,rel	Decrement direct byte and Jump if Not Zero	3	24
NOP		No Operation	1	12

All mnemonics copyrighted ©Intel Corporation 1980



Table 11. Instruction Opcodes in Hexadecimal Order

Hex Code	Number of Bytes	Mnemonic	Operands
00	1	NOP	
01	2	AJMP	code addr
02	3	LJMP	code addr
03	1	RR	A
04	1	INC	Α
05	2	INC	data addr
06	1	INC	@R0
07	1	INC	@R1
08	1	INC	R0
09	1	INC	R1
0A	1	INC	R2
0B	1	INC	R3
OC.	1	INC	R4
0D	1	INC	R5
0E	1	INC	R6
0F	1	INC	R7
10	3	JBC	bit addr, code addr
11	2	ACALL	code addr
12	3	LCALL	code addr
13	1	RRC	A
14	1	DEC	A
15	2	DEC	data addr
16	1	DEC	@R0
17	1	DEC	@R1
18	1	DEC	R0
19	1	DEC	R1
1A	1	DEC	R2
1B	1	DEC	R3
1C	1	DEC	R4
1D	1	DEC	R5
1E	1	DEC	R6
1F	1 3	DEC	R7
20	2	JB	bit addr, code addr
21 22	1	AJMP RET	code addr
23	1	RL	A
24	2	ADD	A, # data
25	2	ADD	A,data addr
26	1	ADD	A,@R0
27	i	ADD	A,@R1
28	i	ADD	A,R0
29	i	ADD	A,R1
2A	i	ADD	A,R2
2B	i	ADD	A,R3
2C	1	ADD	A,R4
2D	1	ADD	A,R5
2E	1	ADD	A,R6
2F	1	ADD	A,R7
30	3	JNB	bit addr, code addr
31	2	ACALL	code addr
32	1	RETI	

Hex Code	Number of Bytes	Mnemonic	Operands
33	1	RLC	A
34	2	ADDC	A.#data
35	2	ADDC	A.data addr
36	1	ADDC	A,@R0
37	1	ADDC	A,@R1
38	1	ADDC	A,R0
39	1	ADDC	A,R1
3A	1	ADDC	A,R2
3B	1	ADDC	A,R3
3C	1	ADDC	A,R4
3D	1	ADDC	A,R5
3E	1	ADDC	A,R6
3F	1	ADDC	A,R7
40	2	JC	code addr
41	2	AJMP	code addr
42	2	ORL	data addr, A
43	3	ORL	data addr, # data
44	2	ORL	A, # data
45	2	ORL	A,data addr
46	1	ORL	A.@R0
47	1	ORL	A,@R1
48	1	ORL	A,R0
49	i	ORL	A,R1
4A	i	ORL	A,R2
4B	1	ORL	A,R3
4C	1	ORL	A,R4
4D	1	ORL	A,R5
4E	1	ORL	A,R6
4F	1	ORL	A,R7
50	2	JNC	code addr
51	2	ACALL	code addr
52	2	ANL	data addr,A
53	3	ANL	data addr, # data
54	2	ANL	A, # data
55	2	ANL	A,data addr
56	1	ANL	A,@R0
57	1	ANL	A,@R1
58	i	ANL	A,R0
59	i	ANL	A,R1
5A	i	ANL	A,R2
5B	i	ANL	A,R3
5C	1	ANL	A,R4
5D	i	ANL	A,R5
5E	1	ANL	A,R6
5E 5F	1	ANL	A.R7
60	2	JZ	code addr
61	2	AJMP	code addr
62	2	XRL	data addr,A
63	3	XRL	data addr, # data
64	2	XRL	A,#data
65	2	XRL	•
00		VUL	A,data addr



Table 11. Instruction Opcodes in Hexadecimal Order (Continued)

Hex Code	Number of Bytes	Mnemonic	Operands
66	1	XRL	A,@R0
67	1	XRL	A,@R1
68	1	XRL	A,R0
69	1	XRL	A,R1
6A	1	XRL	A,R2
6B	1	XRL	A,R3
6C	1	XRL	A,R4
6D	1	XRL	A,R5
6E	1	XRL	A,R6
6F	1	XRL	A,R7
70	2	JNZ	code addr
71	2	ACALL	code addr
72	2	ORL	C,bit addr
73	1	JMP	@A+DPTR
74	2	MOV	A, # data
75	3	MOV	data addr, #data
76	2	MOV	@R0,#data
77	2	MOV	@R1,#data
78	2	MOV	R0, #data
79	2	MOV	R1,#data
7A	2	MOV	R2,#data
7B	2	MOV	R3, # data
7C	2	MOV	R4,#data
7D	2	MOV	R5,#data
7E	2	MOV	R6,#data
7F	2	MOV	R7,#data
80	2	SJMP	code addr
81	2	AJMP	code addr
82	2	ANL	C,bit addr
83	1	MOVC	A,@A + PC
84	1	DIV	AB
85	3	MOV	data addr. data addr
86	2	MOV	data addr.@R0
87	2	MOV	data addr,@R1
88	2	MOV	data addr.R0
89	2	MOV	data addr,R1
8A	2	MOV	data addr,R2
8B	2	MOV	data addr,R3
8C	2	MOV	data addr.R4
8D	2	MOV	data addr,R5
8E	2	MOV	data addr,R6
8F	2	MOV	data addr.R7
90	3	MOV	DPTR,#data
91	2	ACALL	code addr
92	2	MOV	bit addr,C
93	1	MOVC	A,@A+DPTR
94	2	SUBB	A,#data
95	2	SUBB	A,data addr
96	1	SUBB	A.@R0
97	i	SUBB	A,@R1
98	1	SUBB	A,R0
	'		

stadecinial Order (Continued)					
Hex Code	Number of Bytes	Mnemonic	Operands		
99	1	SUBB	A,R1		
9A	1	SUBB	A,R2		
9B	1	SUBB	A,R3		
9C	1	SUBB	A,R4		
9D	1	SUBB	A,R5		
9E	1	SUBB	A,R6		
9F	1	SUBB	A,R7		
A0	2	ORL	C,/bit addr		
A1	2	AJMP	code addr		
A2	2	MOV	C,bit addr		
A3	1	INC	DPTR		
A4	1	MUL	AB		
A5		reserved	Ī		
A6	2	MOV	@R0,data addr		
A7	2	MOV	@R1,data addr		
A8	2	MOV	R0,data addr		
A9	2	MOV	R1,data addr		
AA	2	MOV	R2,data addr		
AB	2	MOV	R3,data addr		
AC	2	MOV	R4,data addr		
AD	2	MOV	R5,data addr		
AE	2	MOV	R6,data addr		
AF	2	MOV	R7,data addr		
B0	2	ANL	C,/bit addr		
B1	2	ACALL	code addr		
B2	2	CPL	bit addr		
B3	1	CPL	C		
B4	3	CJNE	A, #data,code addr		
B5	3	CJNE	A,data addr,code addr		
B6	3	CJNE	@R0, #data,code addr		
B7	3	CJNE	@R1,#data,code addr		
B8	3	CJNE	R0, # data, code addr		
B9	3	CJNE	R1,#data,code addr		
BA	3	CJNE	R2, # data, code addr		
BB	3	CJNE	R3, # data,code addr		
BC	3	CJNE	R4, # data, code addr		
BD	3	CJNE	R5, # data, code addr		
BE	3	CJNE	R6, # data, code addr		
BF	3	CJNE	R7,#data,code addr		
CO	2	PUSH	data addr		
C1	2	AJMP	code addr		
C2	2	CLR	bit addr		
C3	1	CLR SWAP	C A		
C4 CE	2				
C5	_	XCH	A,data addr		
C6	1	XCH	A,@R0		
C7	1	XCH	A,@R1		
C8	1	XCH	A,R0		
C9	1	XCH	A,R1		
CA	1	XCH	A,R2		
CB	1	XCH	A,R3		



Table 11. Instruction Opcodes in Hexadecimal Order (Continued)

		TODIC TITLE	istruction Opcodes	
Hex Code	Number of Bytes	Mnemonic	Operands	
CC	1	XCH	A,R4	
CD	1	XCH	A,R5	
CE	1	XCH	A,R6	
CF	1	XCH	A,R7	
D0	2	POP	data addr	
D1	2	ACALL	code addr	
D2	2	SETB	bit addr	
D3	1	SETB	С	
D4	1	DA	A	
D5	3	DJNZ	data addr,code addr	
D6	1	XCHD	A,@R0	
D7	1	XCHD	A,@R1	
D8	2	DJNZ	R0,code addr	
D9	2	DJNZ	R1,code addr	
DA	2	DJNZ	R2,code addr	
DB	2	DJNZ	R3,code addr	
DC	2	DJNZ	R4,code addr	
DD	2	DJNZ	R5,code addr	
DE	2	DJNZ	R6,code addr	
DF	2	DJNZ	R7,code addr	
E0	1	MOVX	A,@DPTR	
E1	2	AJMP	code addr	
E2	1	MOVX	A,@R0	
E3	1	MOVX	A,@R1	
E4	1	CLR	A	
E5	2	MOV	A,data addr	

Hex	Number	Manage	0
Code	of Bytes	Mnemonic	Operands
E6	1	MOV	A,@R0
E7	1	MOV	A,@R1
E8	1	MOV	A,R0
E9	1	MOV	A,R1
EA	1	MOV	A,R2
EB	1	MOV	A,R3
EC	1	MOV	A,R4
ED	1	MOV	A,R5
EE	1	MOV	A,R6
EF	1	MOV	A,R7
F0	1	MOVX	@DPTR,A
F1	2	ACALL	code addr
F2	1	MOVX	@R0,A
F3	1	MOVX	@R1,A
F4	1	CPL	Α
F5	2	MOV	data addr,A
F6	1	MOV	@R0,A
F7	1	MOV	@R1,A
F8	1	MOV	R0,A
F9	1	MOV	R1,A
FA	1	MOV	R2,A
FB	1	MOV	R3,A
FC	1	MOV	R4,A
FD	1	MOV	R5,A
FE	1	MOV	R6,A
FF	1	MOV	R7,A



INSTRUCTION DEFINITIONS

ACALL addr11

Function: Absolute Call

Description: ACALL unconditionally calls a subroutine located at the indicated address. The instruction

increments the PC twice to obtain the address of the following instruction, then pushes the 16-bit result onto the stack (low-order byte first) and increments the Stack Pointer twice. The destination address is obtained by successively concatenating the five high-order bits of the incremented PC, opcode bits 7-5, and the second byte of the instruction. The subroutine called must therefore start within the same 2K block of the program memory as the first byte of the

instruction following ACALL. No flags are affected.

Example: Initially SP equals 07H. The label "SUBRTN" is at program memory location 0345 H. After

executing the instruction,

ACALL SUBRTN

at location 0123H, SP will contain 09H, internal RAM locations 08H and 09H will contain

25H and 01H, respectively, and the PC will contain 0345H.

Bytes: 2

Cycles: 2

Encoding: a10 a9 a8 1 0 0 0 1

a7 a6 a5 a4 a3 a2 a1 a0

Operation: ACALL

 $(PC) \leftarrow (PC) + 2$ $(SP) \leftarrow (SP) + 1$ $((SP)) \leftarrow (PC_{7-0})$ $(SP) \leftarrow (SP) + 1$ $((SP)) \leftarrow (PC_{15-8})$

 $(PC_{10-0}) \leftarrow page address$

ADD A, < src-byte >

Function:

Add

Description:

ADD adds the byte variable indicated to the Accumulator, leaving the result in the Accumulator. The carry and auxiliary-carry flags are set, respectively, if there is a carry-out from bit 7 or bit 3, and cleared otherwise. When adding unsigned integers, the carry flag indicates an overflow occured.

OV is set if there is a carry-out of bit 6 but not out of bit 7, or a carry-out of bit 7 but not bit 6; otherwise OV is cleared. When adding signed integers, OV indicates a negative number produced as the sum of two positive operands, or a positive sum from two negative operands.

Four source operand addressing modes are allowed: register, direct, register-indirect, or immediate.

Example:

The Accumulator holds 0C3H (11000011B) and register 0 holds 0AAH (10101010B). The instruction,

ADD A.RO

will leave 6DH (01101101B) in the Accumulator with the AC flag cleared and both the carry flag and OV set to 1.

ADD A,Rn

Bytes: 1

Cycles: 1

Encoding: 0010 1 rrr

Operation:

ADD

 $(A) \leftarrow (A) + (Rn)$

ADD A.direct

Bytes: 2

Cycles: 1

Encoding:

0010 0101

direct address

Operation: ADD

 $(A) \leftarrow (A) + (direct)$

ADD A,@Ri

Bytes: 1 Cycles: 1

Encoding:

0010 0111

Operation: ADD

 $(A) \leftarrow (A) + ((R_i))$

ADD A,#data

Bytes: 2 **Cycles:** 1

Encoding:

0010 0100

immediate data

Operation: ADD

 $(A) \leftarrow (A) + \#data$

ADDC A, < src-byte >

Function: Add with Carry

Description: ADDC simultaneously adds the byte variable indicated, the carry flag and the Accumulator

contents, leaving the result in the Accumulator. The carry and auxiliary-carry flags are set, respectively, if there is a carry-out from bit 7 or bit 3, and cleared otherwise. When adding

unsigned integers, the carry flag indicates an overflow occured.

OV is set if there is a carry-out of bit 6 but not out of bit 7, or a carry-out of bit 7 but not out of bit 6; otherwise OV is cleared. When adding signed integers, OV indicates a negative number produced as the sum of two positive operands or a positive sum from two negative operands.

Four source operand addressing modes are allowed: register, direct, register-indirect, or imme-

diate.

Example: The Accumulator holds 0C3H (11000011B) and register 0 holds 0AAH (10101010B) with the

carry flag set. The instruction,

ADDC A,R0

will leave 6EH (01101110B) in the Accumulator with AC cleared and both the Carry flag and

OV set to 1.

ADDC A,Rn

Bytes: 1 Cycles: 1

Encoding: 0 0 1 1 1 rrr

Operation: ADDC

 $(A) \leftarrow (A) + (C) + (R_n)$

ADDC A, direct

Bytes: 2 **Cycles:** 1

Encoding: 0 0 1 1 0 1 0 1 direct address

Operation: ADDC

 $(A) \leftarrow (A) + (C) + (direct)$

ADDC A,@Ri

Bytes: 1 **Cycles:** 1

Encoding: 0 0 1 1 0 1 1 i

Operation: ADDC

 $(A) \leftarrow (A) + (C) + ((R_i))$

ADDC A,#data

Bytes: 2 **Cycles:** 1

Encoding: 0 0 1 1 0 1 0 0 immediate data

Operation: ADDC

 $(A) \leftarrow (A) + (C) + \#data$

AJMP addr11

Function: Absolute Jump

Description: AJMP transfers program execution to the indicated address, which is formed at run-time by

concatenating the high-order five bits of the PC (after incrementing the PC twice), opcode bits 7-5, and the second byte of the instruction. The destination must therefore be within the same

2K block of program memory as the first byte of the instruction following AJMP.

Example: The label "JMPADR" is at program memory location 0123H. The instruction,

AJMP JMPADR

is at location 0345H and will load the PC with 0123H.

Bytes: 2 **Cycles:**

a10 a9 a8 0 **Encoding:** 0001

a7 a6 a5 a4 a3 a2 a1 a0

Operation: **AJMP**

> $(PC) \leftarrow (PC) + 2$ $(PC_{10-0}) \leftarrow page address$

<dest-byte>,<src-byte> ANL

Function: Logical-AND for byte variables

Description: ANL performs the bitwise logical-AND operation between the variables indicated and stores

the results in the destination variable. No flags are affected.

The two operands allow six addressing mode combinations. When the destination is the Accumulator, the source can use register, direct, register-indirect, or immediate addressing; when the destination is a direct address, the source can be the Accumulator or immediate data.

Note: When this instruction is used to modify an output port, the value used as the original

port data will be read from the output data latch, not the input pins.

Example: If the Accumulator holds OC3H (11000011B) and register 0 holds 55H (01010101B) then the

instruction.

ANL A.RO

will leave 41H (01000001B) in the Accumulator.

When the destination is a directly addressed byte, this instruction will clear combinations of bits in any RAM location or hardware register. The mask byte determining the pattern of bits to be cleared would either be a constant contained in the instruction or a value computed in the Accumulator at run-time. The instruction.

ANL P1.#01110011B

will clear bits 7, 3, and 2 of output port 1.

ANL A,Rn

Bytes: 1 Cycles: 1

Encoding: 0 1 0 1 1 rrr

Operation: ANL

 $(A) \leftarrow (A) \land (Rn)$

ANL A, direct

Bytes: 2 **Cycles:** 1

Encoding: 0 1 0 1 0 1 0 1 direct address

Operation: ANL

 $(A) \leftarrow (A) \land (direct)$

ANL A,@Ri

Bytes: 1
Cycles: 1

Encoding: 0 1 0 1 0 1 1 i

Operation: ANL

 $(A) \leftarrow (A) \land ((Ri))$

ANL A, #data

Bytes: 2
Cycles: 1

Encoding: 0 1 0 1 0 0 0 immediate data

Operation: ANL

 $(A) \leftarrow (A) \land \# data$

ANL direct,A

Bytes: 2 **Cycles:** 1

Encoding: 0 1 0 1 0 0 1 0 direct address

Operation: ANL

 $(direct) \leftarrow (direct) \land (A)$

MCS®-51 PROGRAMMER'S GUIDE AND INSTRUCTION SET

ANL direct, # data

Bytes: 3

Cycles: 2

Encoding:

0101 0011

direct address

immediate data

Operation: ANL

(direct) ← (direct) ∧ #data

ANL C, < src-bit >

Function: Logical-AND for bit variables

Description: If the Boolean value of the source bit is a logical 0 then clear the carry flag; otherwise leave the

carry flag in its current state. A slash ("/") preceding the operand in the assembly language indicates that the logical complement of the addressed bit is used as the source value, but the

source bit itself is not affected. No other flags are affected.

Only direct addressing is allowed for the source operand.

Example: Set the carry flag if, and only if, P1.0 = 1, ACC. 7 = 1, and OV = 0:

MOV C,P1.0 ;LOAD CARRY WITH INPUT PIN STATE

ANL C,ACC.7 ;AND CARRY WITH ACCUM. BIT 7

ANL C,/OV ;AND WITH INVERSE OF OVERFLOW FLAG

ANL C,bit

Bytes: 2

Cycles: 2

Encoding: 1 0 0 0 0 0 1 0 bit address

Operation: ANL

 $(C) \leftarrow (C) \land (bit)$

ANL C,/bit

Bytes: 2

Cycles: 2

Encoding: 1 0 1 1 0 0 0 0 bit address

Operation: ANL

 $(C) \leftarrow (C) \land \neg$ (bit)



CJNE <dest-byte>, < src-byte>, rel

Function:

Compare and Jump if Not Equal.

Description:

CJNE compares the magnitudes of the first two operands, and branches if their values are not equal. The branch destination is computed by adding the signed relative-displacement in the last instruction byte to the PC, after incrementing the PC to the start of the next instruction. The carry flag is set if the unsigned integer value of <dest-byte> is less than the unsigned integer value of <src-byte>; otherwise, the carry is cleared. Neither operand is affected.

The first two operands allow four addressing mode combinations: the Accumulator may be compared with any directly addressed byte or immediate data, and any indirect RAM location or working register can be compared with an immediate constant.

Example:

The Accumulator contains 34H. Register 7 contains 56H. The first instruction in the sequence.

```
CJNE R7,#60H, NOT_EQ; R7 = 60H.

NOT_EQ: JC REQ_LOW; IF R7 < 60H.

... ... ; R7 > 60H.
```

sets the carry flag and branches to the instruction at label NOT_EQ. By testing the carry flag, this instruction determines whether R7 is greater or less than 60H.

If the data being presented to Port 1 is also 34H, then the instruction,

WAIT: CJNE A,P1,WAIT

clears the carry flag and continues with the next instruction in sequence, since the Accumulator does equal the data read from P1. (If some other value was being input on P1, the program will loop at this point until the P1 data changes to 34H.)

CJNE A,direct,rel

Bytes: 3 **Cycles:** 2

Encoding:

1011 0101	
-----------	--

 $(C) \leftarrow 0$

direct address

rel. address

Operation:

$$\begin{array}{l} (PC) \longleftarrow (PC) \, + \, 3 \\ \text{IF (A) <> (direct)} \\ \text{THEN} \\ \qquad \qquad (PC) \longleftarrow (PC) \, + \, \textit{relative offset} \\ \text{IF (A) < (direct)} \\ \text{THEN} \\ \qquad \qquad (C) \longleftarrow 1 \\ \text{ELSE} \end{array}$$

CJNE A, # data, rei

Bytes: 3

2

Cycles:

Encoding:

1011 0100

immediate data

rel. address

Operation:

 $(PC) \leftarrow (PC) + 3$ IF (A) <> data

THEN

 $(PC) \leftarrow (PC) + relative offset$

IF(A) < data

THEN

 $(C) \leftarrow 1$

ELSE

 $(C) \leftarrow 0$

CJNE Rn, # data, rel

Bytes:

3 2

Cycles:

Encoding:

1011 1 rrr immediate data

rel. address

Operation:

 $(PC) \leftarrow (PC) + 3$ IF (Rn) <> data

THEN

 $(PC) \leftarrow (PC) + relative offset$

IF(Rn) < data

THEN

 $(C) \leftarrow 1$

ELSE $(C) \leftarrow 0$

CJNE @Ri,#data,rel

Bytes:

2 Cycles:

Encoding:

1011 0 1 1 i

immediate data

rei. address

Operation:

 $(PC) \leftarrow (PC) + 3$

IF ((Ri)) <> data

THEN

 $(PC) \leftarrow (PC) + relative offset$

IF ((Ri)) < data

THEN

 $(C) \leftarrow 1$

ELSE

 $(C) \leftarrow 0$

CLR A

Function:

Clear Accumulator

Description:

The Accumulator is cleared (all bits set on zero). No flags are affected.

Example:

The Accumulator contains 5CH (01011100B). The instruction,

CLR A

will leave the Accumulator set to 00H (0000000B).

Bytes:

1 1

Cycles:

Encoding:

1110 0100

Operation:

CLR $(A) \leftarrow 0$

CLR bit

Function:

Clear bit

Description:

The indicated bit is cleared (reset to zero). No other flags are affected. CLR can operate on the

carry flag or any directly addressable bit.

Example:

Port 1 has previously been written with 5DH (01011101B). The instruction,

CLR P1.2

will leave the port set to 59H (01011001B).

CLR C

Bytes:

Cycles:

Encoding:

1100 0011

Operation:

CLR $(C) \leftarrow 0$

2

1

1

CLR bit

Bytes:

Cycles:

Encoding:

1100 0010 bit address

Operation:

CLR (bit) ← 0

CPL A

Function: Complement Accumulator

Description: Each bit of the Accumulator is logically complemented (one's complement). Bits which previ-

ously contained a one are changed to a zero and vice-versa. No flags are affected.

Example: The Accumulator contains 5CH (01011100B). The instruction,

CPL A

will leave the Accumulator set to 0A3H (10100011B).

Bytes: 1

Cycles: 1

Encoding:

1111 0100

Operation:

CPL(A) $\leftarrow \neg (A)$

CPL bit

Function: Complement bit

Description: The bit variable specified is complemented. A bit which had been a one is changed to zero and

vice-versa. No other flags are affected. CLR can operate on the carry or any directly address-

able bit.

Note: When this instruction is used to modify an output pin, the value used as the original data

will be read from the output data latch, not the input pin.

Example: Port 1 has previously been written with 5BH (01011101B). The instruction sequence,

CPL P1.1

CPL P1.2

will leave the port set to 5BH (01011011B).

CPL C

Bytes: 1

Cycles: 1

Encoding: 1 0 1 1 0 0 1 1

Operation: CPL

(C) ← ¬(C)



MCS®-51 PROGRAMMER'S GUIDE AND INSTRUCTION SET

CPL bit

Bytes: 2

Cycles:

Encoding: 1 0 1 1

bit address

Operation: CPL

(bit) $\leftarrow \neg$ (bit)

DA A

Function:

Decimal-adjust Accumulator for Addition

0010

Description:

DA A adjusts the eight-bit value in the Accumulator resulting from the earlier addition of two variables (each in packed-BCD format), producing two four-bit digits. Any ADD or ADDC instruction may have been used to perform the addition.

If Accumulator bits 3-0 are greater than nine (xxxx1010-xxxx1111), or if the AC flag is one, six is added to the Accumulator producing the proper BCD digit in the low-order nibble. This internal addition would set the carry flag if a carry-out of the low-order four-bit field propagated through all high-order bits, but it would not clear the carry flag otherwise.

If the carry flag is now set, or if the four high-order bits now exceed nine (1010xxxx-111xxxx), these high-order bits are incremented by six, producing the proper BCD digit in the high-order nibble. Again, this would set the carry flag if there was a carry-out of the high-order bits, but wouldn't clear the carry. The carry flag thus indicates if the sum of the original two BCD variables is greater than 100, allowing multiple precision decimal addition. OV is not affected.

All of this occurs during the one instruction cycle. Essentially, this instruction performs the decimal conversion by adding 00H, 06H, 60H, or 66H to the Accumulator, depending on initial Accumulator and PSW conditions.

Note: DA A cannot simply convert a hexadecimal number in the Accumulator to BCD notation, nor does DA A apply to decimal subtraction.



Example:

The Accumulator holds the value 56H (01010110B) representing the packed BCD digits of the decimal number 56. Register 3 contains the value 67H (01100111B) representing the packed BCD digits of the decimal number 67. The carry flag is set. The instruction sequence.

ADDC A,R3 DA A

will first perform a standard twos-complement binary addition, resulting in the value OBEH (10111110) in the Accumulator. The carry and auxiliary carry flags will be cleared.

The Decimal Adjust instruction will then alter the Accumulator to the value 24H (00100100B), indicating the packed BCD digits of the decimal number 24, the low-order two digits of the decimal sum of 56, 67, and the carry-in. The carry flag will be set by the Decimal Adjust instruction, indicating that a decimal overflow occurred. The true sum 56, 67, and 1 is 124.

BCD variables can be incremented or decremented by adding 01H or 99H. If the Accumulator initially holds 30H (representing the digits of 30 decimal), then the instruction sequence,

ADD A, #99H

DA A

will leave the carry set and 29H in the Accumulator, since 30 + 99 = 129. The low-order byte of the sum can be interpreted to mean 30 - 1 = 29.

Bytes: 1

Cycles: 1

Encoding:

1101 0100

Operation:

DA

-contents of Accumulator are BCD IF $[[(A_{3-0}) > 9] \lor [(AC) = 1]]$ THEN $(A_{3-0}) \leftarrow (A_{3-0}) + 6$

AND

IF $[[(A_{7-4}) > 9] \lor [(C) = 1]]$ THEN $(A_{7-4}) \leftarrow (A_{7-4}) + 6$

DEC byte

Function:

Decrement

Description:

The variable indicated is decremented by 1. An original value of 00H will underflow to 0FFH. No flags are affected. Four operand addressing modes are allowed: accumulator, register,

direct, or register-indirect.

Note: When this instruction is used to modify an output port, the value used as the original

port data will be read from the output data latch, not the input pins.

Example:

Register 0 contains 7FH (01111111B). Internal RAM locations 7EH and 7FH contain 00H

and 40H, respectively. The instruction sequence,

DEC @R0

DEC RO

DEC @R0

will leave register 0 set to 7EH and internal RAM locations 7EH and 7FH set to 0FFH and

3FH.

DEC A

Bytes: 1

Cycles: 1

Encoding:

0001 0 1 0 0

Operation:

DEC

 $(A) \leftarrow (A) - 1$

DEC Rn

Bytes: 1

Cycles:

Encoding:

0001 1 rrr

Operation: DEC

 $(Rn) \leftarrow (Rn) - 1$

DEC direct

Bytes: 2 **Cycles:** 1

Encoding: 0

0001 0101

direct address

Operation: DEC

 $(direct) \leftarrow (direct) - 1$

DEC @Ri

Bytes: 1 **Cycles:** 1

Encoding:

0001 011i

Operation: DEC

 $((Ri)) \leftarrow ((Ri)) - 1$

DIV AB

Function: Divide

Description: DIV AB divides the unsigned eight-bit integer in the Accumulator by the unsigned eight-bit

integer in register B. The Accumulator receives the integer part of the quotient; register B

receives the integer remainder. The carry and OV flags will be cleared.

Exception: if B had originally contained 00H, the values returned in the Accumulator and B-register will be undefined and the overflow flag will be set. The carry flag is cleared in any

case.

Example: The Accumulator contains 251 (OFBH or 11111011B) and B contains 18 (12H or 00010010B).

The instruction,

DIV AB

will leave 13 in the Accumulator (0DH or 00001101B) and the value 17 (11H or 00010001B)

in B, since $251 = (13 \times 18) + 17$. Carry and OV will both be cleared.

Bytes: 1

Cycles: 4

Encoding:

1000 0100

Operation: DIV

 $^{(A)}_{(B)_{7.0}}$ \leftarrow $^{(A)/(B)}$



DJNZ <byte>,<rel-addr>

Function:

Decrement and Jump if Not Zero

Description:

DJNZ decrements the location indicated by 1, and branches to the address indicated by the second operand if the resulting value is not zero. An original value of 00H will underflow to 0FFH. No flags are affected. The branch destination would be computed by adding the signed relative-displacement value in the last instruction byte to the PC, after incrementing the PC to the first byte of the following instruction.

The location decremented may be a register or directly addressed byte.

Note: When this instruction is used to modify an output port, the value used as the original port data will be read from the output data latch, not the input pins.

Example:

Internal RAM locations 40H, 50H, and 60H contain the values 01H, 70H, and 15H, respectively. The instruction sequence.

DJNZ 40H,LABEL_1 DJNZ 50H,LABEL_2 DJNZ 60H,LABEL_3

will cause a jump to the instruction at label LABEL_2 with the values 00H, 6FH, and 15H in the three RAM locations. The first jump was *not* taken because the result was zero.

This instruction provides a simple way of executing a program loop a given number of times, or for adding a moderate time delay (from 2 to 512 machine cycles) with a single instruction. The instruction sequence,

17 E.

MOV

R2.#8

TOGGLE:

CPL P1.7

DJNZ

R2.TOGGLE

will toggle P1.7 eight times, causing four output pulses to appear at bit 7 of output Port 1. Each pulse will last three machine cycles; two for DJNZ and one to alter the pin.

DJNZ Rn.rel

Bytes: 2

2

Cycles:

Encodina:

1101 1rrr

rel. address

Operation:

DJNZ $(PC) \leftarrow (PC) + 2$

 $(Rn) \leftarrow (Rn) - 1$ IF (Rn) > 0 or (Rn) < 0

THEN

 $(PC) \leftarrow (PC) + rel$

DJNZ direct.rel

Bytes: 3 Cycles: 2

Encoding:

1101 0101

direct address

rel. address

Operation: DJNZ

 $(PC) \leftarrow (PC) + 2$ $(direct) \leftarrow (direct) - 1$ IF (direct) > 0 or (direct) < 0THEN $(PC) \leftarrow (PC) + rel$

INC <byte>

Function: Increment

Description: INC increments the indicated variable by 1. An original value of 0FFH will overflow to 00H.

No flags are affected. Three addressing modes are allowed: register, direct, or register-indirect.

Note: When this instruction is used to modify an output port, the value used as the original

port data will be read from the output data latch, not the input pins.

Example: Register 0 contains 7EH (011111110B). Internal RAM locations 7EH and 7FH contain 0FFH

and 40H, respectively. The instruction sequence,

INC @R0 INC R0 INC @R0

will leave register 0 set to 7FH and internal RAM locations 7EH and 7FH holding (respective-

ly) 00H and 41H.

INC A

Bytes: 1 Cycles: 1

Encoding: 0 0 0 0 0 1 0 0

Operation: INC

 $(A) \leftarrow (A) + 1$

INC Rn

Bytes: 1
Cycles: 1

Encoding: 0 0

0000 1 rrr

Operation: INC

 $(Rn) \leftarrow (Rn) + 1$

INC direct

Bytes: 2 **Cycles:** 1

Encoding:

0000 0101

direct address

Operation: INC

 $(direct) \leftarrow (direct) + 1$

INC @Ri

Bytes: 1 **Cycles:** 1

Encoding:

0000 011i

Operation: INC

 $((Ri)) \leftarrow ((Ri)) + 1$

INC DPTR

Function: Increment Data Pointer

Description: Increment the 16-bit data pointer by 1. A 16-bit increment (modulo 2¹⁶) is performed; an

overflow of the low-order byte of the data pointer (DPL) from 0FFH to 00H will increment

the high-order byte (DPH). No flags are affected.

This is the only 16-bit register which can be incremented.

Example: Registers DPH and DPL contain 12H and 0FEH, respectively. The instruction sequence,

INC DPTR
INC DPTR
INC DPTR

will change DPH and DPL to 13H and 01H.

Bytes: 1 Cycles: 2

Encoding: 1 0 1 0 0 0 1 1

Operation: INC

 $(DPTR) \leftarrow (DPTR) + 1$

JB bit.rei

Function:

Jump if Bit set

Description:

If the indicated bit is a one, jump to the address indicated; otherwise proceed with the next instruction. The branch destination is computed by adding the signed relative-displacement in the third instruction byte to the PC, after incrementing the PC to the first byte of the next instruction. The bit tested is not modified. No flags are affected.

Example:

The data present at input port 1 is 11001010B. The Accumulator holds 56 (01010110B). The

instruction sequence,

JB P1.2.LABEL1

JB ACC.2,LABEL2

will cause program execution to branch to the instruction at label LABEL2.

Bytes: 3 2 Cycles:

Encoding:

0010 0000

bit address

rel. address

Operation: JB

 $(PC) \leftarrow (PC) + 3$ IF (bit) = 1THEN

 $(PC) \leftarrow (PC) + rel$

JBC bit.rel

Function:

Jump if Bit is set and Clear bit

Description:

If the indicated bit is one, branch to the address indicated; otherwise proceed with the next instruction. The bit will not be cleared if it is already a zero. The branch destination is computed by adding the signed relative-displacement in the third instruction byte to the PC, after incrementing the PC to the first byte of the next instruction. No flags are affected.

Note: When this instruction is used to test an output pin, the value used as the original data

will be read from the output data latch, not the input pin.

Example:

The Accumulator holds 56H (01010110B). The instruction sequence,

JBC ACC.3,LABEL1 JBC ACC.2,LABEL2

will cause program execution to continue at the instruction identified by the label LABEL2, with the Accumulator modified to 52H (01010010B).

Bytes: 3 Cycles: 2

Encoding: 0 0 0 1 0 0 0 0

bit address

rel. address

Operation: JBC

$$(PC) \leftarrow (PC) + 3$$
IF $(bit) = 1$
THEN

(bit)
$$\leftarrow$$
 0
(PC) \leftarrow (PC) + rel

JC rel

Function: Jump if Carry is set

Description: If the carry flag is set, branch to the address indicated; otherwise proceed with the next

instruction. The branch destination is computed by adding the signed relative-displacement in the second instruction byte to the PC, after incrementing the PC twice. No flags are affected.

Example: The carry flag is cleared. The instruction sequence,

JC LABEL1 CPL C JC LABEL 2

will set the carry and cause program execution to continue at the instruction identified by the

label LABEL2.

Bytes: 2 **Cycles:** 2

Encoding: 0 1 0 0 0 0 0

rel. address

Operation: JC

$$(PC) \leftarrow (PC) + 2$$

IF $(C) = 1$
THEN

 $(PC) \leftarrow (PC) + rel$

MCS®-51 PROGRAMMER'S GUIDE AND INSTRUCTION SET

JMP @A+DPTR

Function: Jump indirect

Description: Add the eight-bit unsigned contents of the Accumulator with the sixteen-bit data pointer, and

load the resulting sum to the program counter. This will be the address for subsequent instruction fetches. Sixteen-bit addition is performed (modulo 2¹⁶): a carry-out from the low-order eight bits propagates through the higher-order bits. Neither the Accumulator nor the Data

Pointer is altered. No flags are affected.

Example: An even number from 0 to 6 is in the Accumulator. The following sequence of instructions will

branch to one of four AJMP instructions in a jump table starting at JMP_TBL:

MOV DPTR, #JMP_TBL

JMP @A+DPTR

JMP_TBL: AJMP LABEL0

AJMP LABEL1 AJMP LABEL2 AJMP LABEL3

If the Accumulator equals 04H when starting this sequence, execution will jump to label LABEL2. Remember that AJMP is a two-byte instruction, so the jump instructions start at every other address.

Bytes: 1 **Cycles:** 2

Encoding: 0111 0011

Operation: JMP $(PC) \leftarrow (A) + (DPTR)$

2-48

JNB bit,rel

Function: Jump if Bit Not set

Description: If the indicated bit is a zero, branch to the indicated address; otherwise proceed with the next

instruction. The branch destination is computed by adding the signed relative-displacement in the third instruction byte to the PC, after incrementing the PC to the first byte of the next

instruction. The bit tested is not modified. No flags are affected.

Example: The data present at input port 1 is 11001010B. The Accumulator holds 56H (01010110B). The

instruction sequence,

JNB P1.3,LABEL1 JNB ACC.3,LABEL2

will cause program execution to continue at the instruction at label LABEL2.

Bytes: 3 **Cycles:** 2

Encoding: 0 0 1 1 0 0 0 0

bit address rel. address

Operation: JNB

 $(PC) \leftarrow (PC) + 3$

IF (bit) = 0

THEN (PC) \leftarrow (PC) + rel.

JNC rel

Function: Jump if Carry not set

Description: If the carry flag is a zero, branch to the address indicated; otherwise proceed with the next

instruction. The branch destination is computed by adding the signed relative-displacement in the second instruction byte to the PC, after incrementing the PC twice to point to the next

instruction. The carry flag is not modified.

Example: The carry flag is set. The instruction sequence,

JNC LABEL1 CPL C

JNC LABEL2

will clear the carry and cause program execution to continue at the instruction identified by

the label LABEL2.

Bytes: 2 Cycles: 2

Encoding: 0 1 0 1 0 0 0 0 rel. address

Operation: JNC

 $(PC) \leftarrow (PC) + 2$

IF (C) = 0

THEN $(PC) \leftarrow (PC) + rel$

JNZ rel

Function: Jump if Accumulator Not Zero

Description: If any bit of the Accumulator is a one, branch to the indicated address; otherwise proceed with

the next instruction. The branch destination is computed by adding the signed relative-displacement in the second instruction byte to the PC, after incrementing the PC twice. The

Accumulator is not modified. No flags are affected.

Example: The Accumulator originally holds 00H. The instruction sequence,

JNZ LABEL1 INC A JNZ LABEL2

will set the Accumulator to 01H and continue at label LABEL2.

Bytes: 2 Cycles: 2

Encoding: 0 1 1 1 0 0 0 0 rel. address

Operation: JNZ

 $(PC) \leftarrow (PC) + 2$ IF $(A) \neq 0$

THEN $(PC) \leftarrow (PC) + rel$

JZ rel

Function: Jump if Accumulator Zero

Description: If all bits of the Accumulator are zero, branch to the address indicated; otherwise proceed with

the next instruction. The branch destination is computed by adding the signed relative-displacement in the second instruction byte to the PC, after incrementing the PC twice. The

Accumulator is not modified. No flags are affected.

Example: The Accumulator originally contains 01H. The instruction sequence,

JZ LABEL1 DEC A JZ LABEL2

will change the Accumulator to 00H and cause program execution to continue at the instruc-

tion identified by the label LABEL2.

Bytes: 2 Cycles: 2

Encoding: 0 1 1 0 0 0 0 0 rel. address

Operation: JZ

 $(PC) \leftarrow (PC) + 2$ IF (A) = 0THEN $(PC) \leftarrow (PC) + rel$



LCALL addr16

Function:

Long call

Description:

LCALL calls a subroutine located at the indicated address. The instruction adds three to the program counter to generate the address of the next instruction and then pushes the 16-bit result onto the stack (low byte first), incrementing the Stack Pointer by two. The high-order and low-order bytes of the PC are then loaded, respectively, with the second and third bytes of the LCALL instruction. Program execution continues with the instruction at this address. The subroutine may therefore begin anywhere in the full 64K-byte program memory address space. No flags are affected.

Example:

Initially the Stack Pointer equals 07H. The label "SUBRTN" is assigned to program memory

location 1234H. After executing the instruction,

LCALL SUBRTN

at location 0123H, the Stack Pointer will contain 09H, internal RAM locations 08H and 09H will contain 26H and 01H, and the PC will contain 1234H.

Bytes: 3
Cycles: 2

Encoding:

0001 0010

addr15-addr8

addr7-addr0

Operation:

LCALL $(PC) \leftarrow (PC) + 3$ $(SP) \leftarrow (SP) + 1$ $((SP)) \leftarrow (PC_{7-0})$ $(SP) \leftarrow (SP) + 1$ $((SP)) \leftarrow (PC_{15-8})$ $(PC) \leftarrow add_{15-0}$

LJMP addr16

Function:

Long Jump

Description:

LJMP causes an unconditional branch to the indicated address, by loading the high-order and low-order bytes of the PC (respectively) with the second and third instruction bytes. The destination may therefore be anywhere in the full 64K program memory address space. No loss are affected

flags are affected.

Example:

The label "JMPADR" is assigned to the instruction at program memory location 1234H. The

instruction,

LJMP JMPADR

at location 0123H will load the program counter with 1234H.

Bytes: 3 **Cycles:** 2

Encoding:

0000 0010

addr15-addr8

addr7-addr0

Operation:

LJMP

 $(PC) \leftarrow addr_{15-0}$

MOV <dest-byte>,<src-byte>

Function: Ma

Move byte variable

Description:

The byte variable indicated by the second operand is copied into the location specified by the first operand. The source byte is not affected. No other register or flag is affected.

This is by far the most flexible operation. Fifteen combinations of source and destination addressing modes are allowed.

Example:

Internal RAM location 30H holds 40H. The value of RAM location 40H is 10H. The data present at input port 1 is 11001010B (OCAH).

MOV R0,#30H ;R0 <= 30H MOV A,@R0 ;A <= 40H MOV R1,A ;R1 <= 40H MOV B,@R1 ;B <= 10H

MOV @R1,P1 ;RAM $(40H) \le 0$ CAH

MOV P2,P1 ;P2 #0CAH

leaves the value 30H in register 0, 40H in both the Accumulator and register 1, 10H in register B, and 0CAH (11001010B) both in RAM location 40H and output on port 2.

MOV A.Rn

Bytes: 1
Cycles: 1

Encoding: 1110 1 rrr

Operation: MOV

 $(A) \leftarrow (Rn)$

*MOV A, direct

Bytes: 2 Cycles: 1

Encoding: 1 1

1110 0101

direct address

Operation: MOV

 $(A) \leftarrow (direct)$

MOV A,ACC is not a valid instruction.

MOV A,@Ri

Bytes: 1. **Cycles:** 1

Encoding:

1110 0 1 1 i

MOV Operation:

 $(A) \leftarrow ((Ri))$

MOV A, # data

Bytes: 2 Cycles: 1

Encoding:

0111 0 1 0 0 immediate data

Operation: MOV

(A) ← #data

MOV Rn,A

Bytes: 1 Cycles:

Encoding:

1111 1 rrr

Operation:

 $(Rn) \leftarrow (A)$

MOV

MOV Rn, direct

Bytes: 2 **Cycles:** 2

Encoding:

1010 1 rrr

direct addr.

Operation: MOV

 $(Rn) \leftarrow (direct)$

MOV Rn, # data

Bytes: 2 Cycles:

Encoding:

0111 1 rrr immediate data

MOV Operation:

(Rn) ← #data

MOV direct.A

MCS®-51 PROGRAMMER'S GUIDE AND INSTRUCTION SET

Bytes: 2 Cycles: 1 1111 **Encoding:** 0 1 0 1 direct address Operation: MOV $(direct) \leftarrow (A)$ MOV direct,Rn **Bytes:** 2 Cycles: **Encoding:** 1000 1 rrr direct address Operation: MOV $(direct) \leftarrow (Rn)$ MOV direct, direct **Bytes:** 3 Cycles: 2 **Encoding:** 1000 0 1 0 1 dir. addr. (src) dir. addr. (dest) Operation: MOV $(direct) \leftarrow (direct)$ MOV direct,@Ri Bytes: 2 Cycles: **Encoding:** 1000 0 1 1 i direct addr. Operation: MOV $(direct) \leftarrow ((Ri))$ MOV direct, # data Bytes: 3 Cycles: 2 **Encoding:** 0 1 1 1 direct address immediate data 0101 Operation: MOV (direct) ← #data

MOV @Ri.A

Bytes: 1 Cycles: 1

Encoding: 1 1 1

1111 011i

Operation: MOV

 $((Ri)) \leftarrow (A)$

MOV @Ri,direct

Bytes: 2 **Cycles:** 2

Encoding:

1010 011i

direct addr.

Operation: MOV

 $((Ri)) \leftarrow (direct)$

MOV @Ri,#data

Bytes: 2 Cycles: 1

Encoding:

0111 011i

immediate data

Operation: MOV

 $((RI)) \leftarrow \# data$

MOV <dest-bit>,<src-bit>

Function: Move bit data

Description: The Boolean variable indicated by the second operand is copied into the location specified by

the first operand. One of the operands must be the carry flag; the other may be any directly

addressable bit. No other register or flag is affected.

Example: The carry flag is originally set. The data present at input Port 3 is 11000101B. The data

previously written to output Port 1 is 35H (00110101B).

MOV P1.3,C MOV C,P3.3 MOV P1.2,C

will leave the carry cleared and change Port 1 to 39H (00111001B).

MOV C,bit

Bytes:

Cycles: 1

2

2

Encoding:

1010 0010

bit address

Operation: MOV

(C) ← (bit)

MOV bit,C

Bytes:

Cycles: 2

Encoding:

1001 0010

bit address

Operation:

MOV (bit) \leftarrow (C)

MOV DPTR,#data16

Function: Load Data Pointer with a 16-bit constant

Description: The Data Pointer is loaded with the 16-bit constant indicated. The 16-bit constant is loaded

into the second and third bytes of the instruction. The second byte (DPH) is the high-order

byte, while the third byte (DPL) holds the low-order byte. No flags are affected.

This is the only instruction which moves 16 bits of data at once.

Example: The instruction,

MOV DPTR, #1234H

will load the value 1234H into the Data Pointer: DPH will hold 12H and DPL will hold 34H.

Bytes: 3 Cycles: 2

Encoding:

1001 0000

immed. data15-8

immed. data7-0

Operation: MOV

 $(DPTR) \leftarrow \#data_{15-0}$

DPH □ DPL ← #data₁₅₋₈ □ #data₇₋₀

MOVC A,@A + < base-reg >

Function:

Move Code byte

Description:

The MOVC instructions load the Accumulator with a code byte, or constant from program memory. The address of the byte fetched is the sum of the original unsigned eight-bit Accumulator contents and the contents of a sixteen-bit base register, which may be either the Data Pointer or the PC. In the latter case, the PC is incremented to the address of the following instruction before being added with the Accumulator; otherwise the base register is not altered. Sixteen-bit addition is performed so a carry-out from the low-order eight bits may propagate through higher-order bits. No flags are affected.

Example:

A value between 0 and 3 is in the Accumulator. The following instructions will translate the value in the Accumulator to one of four values defined by the DB (define byte) directive.

REL_PC: INC A

MOVC A,@A+PC

RET

DB 66H

DB 77H

DB 88H

DB 99H

If the subroutine is called with the Accumulator equal to 01H, it will return with 77H in the Accumulator. The INC A before the MOVC instruction is needed to "get around" the RET instruction above the table. If several bytes of code separated the MOVC from the table, the corresponding number would be added to the Accumulator instead.

MOVC A,@A+DPTR

Bytes:

Cycles: 2

Encoding: 1 0 0 1

1

Operation: MOVC

 $(A) \leftarrow ((A) + (DPTR))$

0011

MOVC A,@A + PC

Bytes: 1
Cycles: 2

Encoding:

1000 0011

Operation: MOVC

 $(PC) \leftarrow (PC) + 1$ $(A) \leftarrow ((A) + (PC))$



MOVX <dest-byte>,<src-byte>

Function:

Move External

Description:

The MOVX instructions transfer data between the Accumulator and a byte of external data memory, hence the "X" appended to MOV. There are two types of instructions, differing in whether they provide an eight-bit or sixteen-bit indirect address to the external data RAM.

In the first type, the contents of R0 or R1 in the current register bank provide an eight-bit address multiplexed with data on P0. Eight bits are sufficient for external I/O expansion decoding or for a relatively small RAM array. For somewhat larger arrays, any output port pins can be used to output higher-order address bits. These pins would be controlled by an output instruction preceding the MOVX.

In the second type of MOVX instruction, the Data Pointer generates a sixteen-bit address. P2 outputs the high-order eight address bits (the contents of DPH) while P0 multiplexes the low-order eight bits (DPL) with data. The P2 Special Function Register retains its previous contents while the P2 output buffers are emitting the contents of DPH. This form is faster and more efficient when accessing very large data arrays (up to 64K bytes), since no additional instructions are needed to set up the output ports.

It is possible in some situations to mix the two MOVX types. A large RAM array with its high-order address lines driven by P2 can be addressed via the Data Pointer, or with code to output high-order address bits to P2 followed by a MOVX instruction using R0 or R1.

Example:

An external 256 byte RAM using multiplexed address/data lines (e.g., an Intel 8155 RAM/I/O/Timer) is connected to the 8051 Port 0. Port 3 provides control lines for the external RAM. Ports 1 and 2 are used for normal I/O. Registers 0 and 1 contain 12H and 34H. Location 34H of the external RAM holds the value 56H. The instruction sequence,

MOVX A,@R1

MOVX @RO,A

copies the value 56H into both the Accumulator and external RAM location 12H.

intel.

MOVX A,@Ri

Bytes: 1

Cycles: 2

Encoding:

1110 0011

Operation:

MOVX $(A) \leftarrow ((Ri))$

MOVX A,@DPTR

Bytes: 1

2

Cycles:

Encoding:

1110 0000

Operation: MOVX

 $(A) \leftarrow ((DPTR))$

MOVX @Ri,A

Bytes: 1

Cycles: 2

Encoding:

1111 001i

Operation: MOVX

 $((Ri)) \leftarrow (A)$

MOVX @DPTR,A

Bytes: 1

Cycles: 2

Encoding:

1111 0000

Operation:

MOVX

 $(DPTR) \leftarrow (A)$

MUL AB

Function:

Multiply

Description:

MUL AB multiplies the unsigned eight-bit integers in the Accumulator and register B. The low-order byte of the sixteen-bit product is left in the Accumulator, and the high-order byte in B. If the product is greater than 255 (OFFH) the overflow flag is set; otherwise it is cleared.

The carry flag is always cleared.

Example:

Originally the Accumulator holds the value 80 (50H). Register B holds the value 160 (0A0H).

The instruction.

MUL AB

will give the product 12,800 (3200H), so B is changed to 32H (00110010B) and the Accumula-

tor is cleared. The overflow flag is set, carry is cleared.

Bytes:

1

Cycles:

4

Encoding:

1010 0100

Operation:

MUL

 $(A)_{7-0} \leftarrow (A) \times (B)$

(B)15-8

NOP

Function:

No Operation

Description:

Execution continues at the following instruction. Other than the PC, no registers or flags are

affected.

Example:

It is desired to produce a low-going output pulse on bit 7 of Port 2 lasting exactly 5 cycles. A simple SETB/CLR sequence would generate a one-cycle pulse, so four additional cycles must be inserted. This may be done (assuming no interrupts are enabled) with the instruction

sequence,

CLR P2.7

NOP

NOP

NOP

NOP

SETB P2.7

Bytes:

1

Cycles:

1

Encoding:

0000 0000

Operation:

NOP

 $(PC) \leftarrow (PC) + 1$

ORL <dest-byte> <src-byte>

Function:

Logical-OR for byte variables

Description:

ORL performs the bitwise logical-OR operation between the indicated variables, storing the results in the destination byte. No flags are affected.

The two operands allow six addressing mode combinations. When the destination is the Accumulator, the source can use register, direct, register-indirect, or immediate addressing; when the destination is a direct address, the source can be the Accumulator or immediate data.

Note: When this instruction is used to modify an output port, the value used as the original port data will be read from the output data latch, not the input pins.

Example:

If the Accumulator holds 0C3H (11000011B) and R0 holds 55H (01010101B) then the instruction.

ORL A.RO

will leave the Accumulator holding the value 0D7H (11010111B).

When the destination is a directly addressed byte, the instruction can set combinations of bits in any RAM location or hardware register. The pattern of bits to be set is determined by a mask byte, which may be either a constant data value in the instruction or a variable computed in the Accumulator at run-time. The instruction,

ORL P1,#00110010B

will set bits 5, 4, and 1 of output Port 1.

ORL A.Rn

Bytes: 1

Cycles: 1

Encoding:

0100 1 rrr

Operation: ORL

 $(A) \leftarrow (A) \lor (Rn)$

ORL A, direct

Bytes: 2 Cycles: 1

Encoding:

0100 0101

direct address

Operation: ORL

 $(A) \leftarrow (A) \lor (direct)$

ORL A,@Ri

Bytes: 1 Cycles: 1

Encoding:

0100 011i

Operation:

 $\begin{array}{l}
ORL \\
(A) \leftarrow (A) \lor ((Ri))
\end{array}$

ORL A,#data

Bytes: 2 **Cycles:** 1

Encoding:

0100 0100

immediate data

Operation:

ORL

 $(A) \leftarrow (A) \lor \#data$

ORL direct,A

Bytes: 2 **Cycles:** 1

Encoding:

0100 0010

direct address

Operation:

 $(direct) \leftarrow (direct) \lor (A)$

ORL direct, # data

Bytes: 3
Cycles: 2

Encoding:

0100 0011

direct addr.

immediate data

Operation: ORL

(direct) ← (direct) ∨ #data

ORL C, < src-bit>

Function: Logical-OR for bit variables

Description: Set the carry flag if the Boolean value is a logical 1; leave the carry in its current state

otherwise. A slash ("/") preceding the operand in the assembly language indicates that the logical complement of the addressed bit is used as the source value, but the source bit itself is

not affected. No other flags are affected.

Example: Set the carry flag if and only if P1.0 = 1, ACC. 7 = 1, or OV = 0:

MOV C,P1.0 ;LOAD CARRY WITH INPUT PIN P10

ORL C,ACC.7 ;OR CARRY WITH THE ACC. BIT 7

ORL C,/OV ;OR CARRY WITH THE INVERSE OF OV.

ORL C,bit

Bytes: 2

Cycles: 2

Encoding: 0 1 1 1 0 0 1 0

bit address

Operation: ORL

 $(C) \leftarrow (C) \lor (bit)$

ORL C,/bit

Bytes: 2 Cycles: 2

Encoding: 1 0 1 0 0 0 0 0

bit address

Operation: ORL

 $(C) \leftarrow (C) \lor (\overline{bit})$

POP direct

Function:

Pop from stack.

Description:

The contents of the internal RAM location addressed by the Stack Pointer is read, and the Stack Pointer is decremented by one. The value read is then transferred to the directly ad-

dressed byte indicated. No flags are affected.

Example:

The Stack Pointer originally contains the value 32H, and internal RAM locations 30H through 32H contain the values 20H, 23H, and 01H, respectively. The instruction sequence,

POP DPH

POP DPL

will leave the Stack Pointer equal to the value 30H and the Data Pointer set to 0123H. At this point the instruction,

POP SP

will leave the Stack Pointer set to 20H. Note that in this special case the Stack Pointer was decremented to 2FH before being loaded with the value popped (20H).

Bytes: Cycles:

Encoding:

1101 0000

direct address

Operation: POP

 $(direct) \leftarrow ((SP))$ $(SP) \leftarrow (SP) - 1$

PUSH direct

Function:

Push onto stack

Description:

The Stack Pointer is incremented by one. The contents of the indicated variable is then copied into the internal RAM location addressed by the Stack Pointer. Otherwise no flags are affect-

ed.

Example:

On entering an interrupt routine the Stack Pointer contains 09H. The Data Pointer holds the

value 0123H. The instruction sequence,

PUSH DPL

PUSH DPH

will leave the Stack Pointer set to 0BH and store 23H and 01H in internal RAM locations 0AH and 0BH, respectively.

2 Bytes: 2 Cycles:

Encoding:

1100 0000

direct address

Operation:

PUSH

 $(SP) \leftarrow (SP) + 1$

 $((SP)) \leftarrow (direct)$

RET

Function: Return from subroutine

Description: RET pops the high- and low-order bytes of the PC successively from the stack, decrementing

the Stack Pointer by two. Program execution continues at the resulting address, generally the

instruction immediately following an ACALL or LCALL. No flags are affected.

Example: The Stack Pointer originally contains the value 0BH. Internal RAM locations 0AH and 0BH

contain the values 23H and 01H, respectively. The instruction,

RET

will leave the Stack Pointer equal to the value 09H. Program execution will continue at

location 0123H.

Bytes: 1 Cycles: 2

Encoding: 0 0 1 0 0 0 1 0

Operation: RET

 $(PC_{15-8}) \leftarrow ((SP))$ $(SP) \leftarrow (SP) - 1$ $(PC_{7-0}) \leftarrow ((SP))$ $(SP) \leftarrow (SP) - 1$

RETI

Function: Return from interrupt

Description: RETI pops the high- and low-order bytes of the PC successively from the stack, and restores

the interrupt logic to accept additional interrupts at the same priority level as the one just processed. The Stack Pointer is left decremented by two. No other registers are affected; the PSW is not automatically restored to its pre-interrupt status. Program execution continues at the resulting address, which is generally the instruction immediately after the point at which the interrupt request was detected. If a lower- or same-level interrupt had been pending when the RETI instruction is executed, that one instruction will be executed before the pending

interrupt is processed.

Example: The Stack Pointer originally contains the value 0BH. An interrupt was detected during the

instruction ending at location 0122H. Internal RAM locations 0AH and 0BH contain the

values 23H and 01H, respectively. The instruction,

RETI

will leave the Stack Pointer equal to 09H and return program execution to location 0123H.

Bytes: 1 Cycles: 2

Encoding: 0 0 1 1 0 0 1 0

Operation: RETI

 $(PC_{15-8}) \leftarrow ((SP))$ $(SP) \leftarrow (SP) - 1$ $(PC_{7-0}) \leftarrow ((SP))$ $(SP) \leftarrow (SP) - 1$

RL A

Function: Rotate Accumulator Left

Description: The eight bits in the Accumulator are rotated one bit to the left. Bit 7 is rotated into the bit 0

position. No flags are affected.

Example: The Accumulator holds the value OC5H (11000101B). The instruction,

RL A

leaves the Accumulator holding the value 8BH (10001011B) with the carry unaffected.

Bytes: 1 **Cycles:** 1

Encoding: 0 0 1 0 0 0 1 1

Operation: RI

 $(A_n + 1) \leftarrow (A_n) \quad n = 0 - 6$

 $(A\ddot{0}) \leftarrow (A7)$

RLC A

Function: Rotate Accumulator Left through the Carry flag

Description: The eight bits in the Accumulator and the carry flag are together rotated one bit to the left. Bit

7 moves into the carry flag; the original state of the carry flag moves into the bit 0 position. No

other flags are affected.

Example: The Accumulator holds the value OC5H (11000101B), and the carry is zero. The instruction,

RLC A

leaves the Accumulator holding the value 8BH (10001010B) with the carry set.

Bytes: 1 **Cycles:** 1

Encoding: 0 0 1 1 0 0 1 1

Operation: RLC

 $(An + 1) \leftarrow (An)$ n = 0 - 6 $(A0) \leftarrow (C)$

 $(A0) \leftarrow (C)$ $(C) \leftarrow (A7)$

RR A

Function: Rotate Accumulator Right

The eight bits in the Accumulator are rotated one bit to the right. Bit 0 is rotated into the bit 7 **Description:**

position. No flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B). The instruction,

RR A

leaves the Accumulator holding the value 0E2H (11100010B) with the carry unaffected.

Bytes: 1 **Cycles:** 1

Encoding: 0000 0011

Operation:

 $(An) \leftarrow (A_n + 1)$ n = 0 - 6

 $(A7) \leftarrow (A0)$

RRC A

Function: Rotate Accumulator Right through Carry flag

Description: The eight bits in the Accumulator and the carry flag are together rotated one bit to the right.

Bit 0 moves into the carry flag; the original value of the carry flag moves into the bit 7

position. No other flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B), the carry is zero. The instruction,

RRC A

leaves the Accumulator holding the value 62 (01100010B) with the carry set.

Bytes: **Cycles:**

0001 0011 **Encoding:**

Operation: RRC

 $(An) \leftarrow (An + 1)$ n = 0 - 6 $(A7) \leftarrow (C)$

 $(C) \leftarrow (A0)$

SETB <bit>

Function: Set Bit

Description: SETB sets the indicated bit to one. SETB can operate on the carry flag or any directly

addressable bit. No other flags are affected.

Example: The carry flag is cleared. Output Port 1 has been written with the value 34H (00110100B). The

instructions,

SETB C

SETB P1.0

will leave the carry flag set to 1 and change the data output on Port 1 to 35H (00110101B).

SETB C

Bytes: 1

Cycles: 1

Encoding: | 1 1 0 1 | 0 0 1 1

Operation: SETB

 $(C) \leftarrow 1$

SETB bit

Bytes: 2 **Cycles:** 1

Encoding: | 1 1 0 1 | 0 0 1 0

bit address

Operation: SETB

(bit) \leftarrow 1

SJMP rei

Function: Short Jump

Description: Program control branches unconditionally to the address indicated. The branch destination is

computed by adding the signed displacement in the second instruction byte to the PC, after incrementing the PC twice. Therefore, the range of destinations allowed is from 128 bytes

preceding this instruction to 127 bytes following it.

Example: The label "RELADR" is assigned to an instruction at program memory location 0123H. The

instruction.

SJMP RELADR

will assemble into location 0100H. After the instruction is executed, the PC will contain the

value 0123H.

(Note: Under the above conditions the instruction following SJMP will be at 102H. Therefore, the displacement byte of the instruction will be the relative offset (0123H-0102H) = 21H. Put another way, an SJMP with a displacement of 0FEH would be a one-instruction infinite loop.)

Bytes: 2 **Cycles:** 2

Encoding: 1 0 0 0 0 0 0 0

rel. address

Operation: SJMP

 $(PC) \leftarrow (PC) + 2$ $(PC) \leftarrow (PC) + rel$

SUBB A, < src-byte >

Function:

Subtract with borrow

Description:

SUBB subtracts the indicated variable and the carry flag together from the Accumulator, leaving the result in the Accumulator. SUBB sets the carry (borrow) flag if a borrow is needed for bit 7, and clears C otherwise. (If C was set before executing a SUBB instruction, this indicates that a borrow was needed for the previous step in a multiple precision subtraction, so the carry is subtracted from the Accumulator along with the source operand.) AC is set if a borrow is needed for bit 3, and cleared otherwise. OV is set if a borrow is needed into bit 6, but not into bit 7, or into bit 7, but not bit 6.

When subtracting signed integers OV indicates a negative number produced when a negative value is subtracted from a positive value, or a positive result when a positive number is subtracted from a negative number.

The source operand allows four addressing modes: register, direct, register-indirect, or immediate.

Example:

The Accumulator holds 0C9H (11001001B), register 2 holds 54H (01010100B), and the carry flag is set. The instruction,

SUBB A.R2

will leave the value 74H (01110100B) in the accumulator, with the carry flag and AC cleared but OV set.

Notice that 0C9H minus 54H is 75H. The difference between this and the above result is due to the carry (borrow) flag being set before the operation. If the state of the carry is not known before starting a single or multiple-precision subtraction, it should be explicitly cleared by a CLR C instruction.

SUBB A.Rn

Bytes: 1
Cycles: 1

Encoding: 1001 1 rrr

Operation: SUBB

 $(A) \leftarrow (A) - (C) - (Rn)$

SUBB A, direct

Bytes: 2 1

Cycles:

Encoding: 1001 0101

direct address

Operation: **SUBB**

 $(A) \leftarrow (A) - (C) - (direct)$

SUBB A,@Ri

1 Bytes: Cycles: 1

Encoding:

1001 0 1 1 i

Operation:

SUBB $(A) \leftarrow (A) - (C) - ((Ri))$

SUBB A,#data

Bytes: 2 **Cycles:** 1

Encoding:

1001 0100

SUBB

immediate data

Operation:

 $(A) \leftarrow (A) - (C) - \#data$

SWAP Α

Function: Swap nibbles within the Accumulator

Description: SWAP A interchanges the low- and high-order nibbles (four-bit fields) of the Accumulator

(bits 3-0 and bits 7-4). The operation can also be thought of as a four-bit rotate instruction. No

flags are affected.

Example: The Accumulator holds the value OC5H (11000101B). The instruction,

SWAP A

leaves the Accumulator holding the value 5CH (01011100B).

1 Bytes: 1 Cycles:

1100 0 1 0 0 **Encoding:**

Operation: **SWAP**

 $(A_{3-0}) \stackrel{\rightarrow}{\leftarrow} (A_{7-4})$

XCH A, < byte>

Function: Exchange Accumulator with byte variable

Description: XCH loads the Accumulator with the contents of the indicated variable, at the same time

writing the original Accumulator contents to the indicated variable. The source/destination

operand can use register, direct, or register-indirect addressing.

Example: R0 contains the address 20H. The Accumulator holds the value 3FH (00111111B). Internal

RAM location 20H holds the value 75H (01110101B). The instruction,

XCH A,@R0

will leave RAM location 20H holding the values 3FH (00111111B) and 75H (01110101B) in

the accumulator.

XCH A,Rn

Bytes: 1 Cycles: 1

Encoding: 1 1 0 0 1 r r r

Operation: XCH

XCH A,direct

Bytes: 2 **Cycles:** 1

Encoding: 1 1 0 0 0 1 0 1 direct address

Operation: XCH

(A) → (direct)

XCH A,@Ri

Bytes: 1 Cycles: 1

Encoding: 1 1 0 0 0 1 1 i

Operation: XCH

XCHD A.@Ri

Function:

Exchange Digit

Description:

XCHD exchanges the low-order nibble of the Accumulator (bits 3-0), generally representing a hexadecimal or BCD digit, with that of the internal RAM location indirectly addressed by the specified register. The high-order nibbles (bits 7-4) of each register are not affected. No flags

are affected.

Example:

R0 contains the address 20H. The Accumulator holds the value 36H (00110110B). Internal

RAM location 20H holds the value 75H (01110101B). The instruction,

XCHD A,@R0

will leave RAM location 20H holding the value 76H (01110110B) and 35H (00110101B) in the

Accumulator.

Bytes:

Cycles:

Encoding:

1101 0 1 1 i

Operation:

XCHD

 $(A_{3-0}) \stackrel{\rightarrow}{=} ((Ri_{3-0}))$

<dest-byte>,<src-byte> XRL

Function:

Logical Exclusive-OR for byte variables

Description:

XRL performs the bitwise logical Exclusive-OR operation between the indicated variables, storing the results in the destination. No flags are affected.

The two operands allow six addressing mode combinations. When the destination is the Accumulator, the source can use register, direct, register-indirect, or immediate addressing; when the destination is a direct address, the source can be the Accumulator or immediate data.

(Note: When this instruction is used to modify an output port, the value used as the original port data will be read from the output data latch, not the input pins.)

Example:

If the Accumulator holds 0C3H (11000011B) and register 0 holds 0AAH (10101010B) then

the instruction,

XRL A.RO

will leave the Accumulator holding the value 69H (01101001B).

When the destination is a directly addressed byte, this instruction can complement combinations of bits in any RAM location or hardware register. The pattern of bits to be complemented is then determined by a mask byte, either a constant contained in the instruction or a variable computed in the Accumulator at run-time. The instruction,

XRL P1,#00110001B

will complement bits 5, 4, and 0 of output Port 1.

XRL A,Rn

Bytes: 1
Cycles: 1

Encoding: 0 1 1 0 1 r r r

Operation: XRL

 $(A) \leftarrow (A) \lor (Rn)$

XRL A,direct

Bytes: 2 **Cycles:** 1

Encoding: 0 1 1 0 0 1 0 1 direct address

Operation: XRL

 $(A) \leftarrow (A) \lor (direct)$

XRL A,@Ri

Bytes: 1
Cycles: 1

Encoding: 0 1 1 0 0 1 1 i

Operation: XRL

 $(A) \leftarrow (A) \lor ((Ri))$

XRL A,#data

Bytes: 2 Cycles: 1

Encoding: 0 1 1 0 0 1 0 0 immediate data

Operation: XRL

 $(A) \leftarrow (A) \ \forall \ \#data$

XRL direct,A

Bytes: 2 **Cycles:** 1

Encoding: 0 1 1 0 0 0 1 0 direct address

Operation: XRL

 $(direct) \leftarrow (direct) \lor (A)$

XRL direct,#data

Bytes: 3 **Cycles:** 2

Encoding:

0110 0011

direct address

immediate data

Operation: XRL

(direct) ← (direct) ¥ #data